

R. Brigola, TH Nürnberg Georg Simon Ohm, 2014

Mathematica - Notebooks als Bonusmaterial zum Lehrbuch

**[1] Rolf Brigola Fourier-Analysis und Distributionen,
Eine Einführung mit Anwendungen,
edition swk, Hamburg 2013**

Demo-Material zur Fouriertransformation und ihren Eigenschaften

Referenzen zu Kapiteln, Abschnitten, Seitenzahlen beziehen sich im Folgenden meist auf das genannte Lehrbuch. Einige wenige weitere Referenzen findet man am Ende des Notebooks.

Ziel diese Notebooks ist, Nutzern wie Teilnehmerinnen in meinen Lehrveranstaltungen einige Beispiele zur Fouriertransformation und ihren Eigenschaften mit Hilfe von *Mathematica* vorzustellen. Ich gehe davon aus, dass Leser grundlegende Eigenschaften der Fouriertransformation parat haben, entweder aus [1], Kap 9 oder anderer verlässlicher Quelle.

Das Notebook ist mit *Mathematica* 10 unter Windows 7 erstellt. Dieses und weitere Demo-Notebooks findet man unter folgender URL des Autors: www.stiftung-swk.de/mathematica

1. Ein paar Funktionen zur nachfolgenden Veranschaulichung von Eigenschaften der Fouriertransformation und ihrer Inversen


Wir definieren zum Start einige wenige Funktionen, mit denen wir dann Eigenschaften der Fouriertransformation (in der Folge abgekürzt FT) demonstrieren wollen.

```
In[1]:= ClearAll["Global`*"]; Remove["Global`*"]; ?Global`*  
Remove::rmnsm : There are no symbols matching "Global`*". >>  
Information::nomatch : No symbol matching Global`* found. >>
```

Beispiel 1. Die Rechteckfunktion $r[a,t]$ mit Träger in $[-a,a]$

```
In[2]:= r[a_, t_] := UnitStep[t + a] - UnitStep[t - a] (* r[1,t] *)
Plot[r[1, t], {t, -2, 3}, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
```


Out[3]=



Beispiel 2. Eine Dreieckfunktion $d[a,t]$ mit Träger in $[-a,a]$


```
In[4]:= d[a_, t_] :=
  (t + a) / a (UnitStep[t + a] - UnitStep[t]) + (a - t) / a (UnitStep[t] - UnitStep[t - a])
Plot[d[2, t], {t, -4, 4}, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
```

Out[5]=



Beispiel 3. Die Gaußfunktion $\text{gauss}[t] = e^{-t^2/2}$ für reelle t

```
In[6]:= gauss[t_] := Exp[-t^2/2];
Plot[gauss[t], {t, -4, 4}, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
```



2. Beispiele zu Fouriertransformation von Funktionen und Distributionen aus S' über \mathbb{R} als Parameterraum

Ich verwende nachfolgend die Fouriertransformation mit der *Mathematica*-Einstellung **FourierParameters**->**{1,-1}**, wie ich es in [1] getan habe. Schreibe ich $F(f)$ für die FT einer Funktion oder Distribution f , dann notiert $F^{-1}(Ff)$ die inverse FourierTransformation,

$$F(f)(\omega) = \int_{-\infty}^{+\infty} f(t) e^{-i\omega t} dt \quad \text{und} \quad F^{-1}(Ff)(t) = \frac{1}{2\pi} \int_{-\infty}^{+\infty} Ff(\omega) e^{i\omega t} d\omega.$$

Wir wollen mit Blick auf Anwendungen insbesondere auch die FT von Distributionen aus S' zeigen. Statt "Fouriertransformierte einer Funktion f " sage ich auch nur kurz "Spektrum von f ". Den Parameterraum für t nenne ich "**Zeitbereich**", den für die FT mit Parameter ω nenne ich "**Frequenzbereich**", da ω in Anwendungen oft für eine Kreisfrequenz steht.

Die für Themen meiner *Mathematica*-Notebook-Serie - die vor allem für Zwecke in der Lehre für Ingenieurstudenten dienen soll - wichtigsten Distributionen sind die Diracsche Delta-Distribution, lokal-integrierbare Funktionen, rationale Funktionen, Impulsfolgen und verallgemeinerte Fourierreihen, die allesamt zu S' gehören.

Ich notiere die Anwendung einer Distribution T auf eine Testfunktion ϕ wie weithin üblich durch $\langle T, \phi \rangle$ oder, um einen Parameter aus dem zugrundeliegenden Parameterraum wie etwa einen Zeitparameter t oder einen Kreisfrequenzparameter ω mit anzugeben, durch $\langle T(t), \phi(t) \rangle$ oder $\langle T(\omega), \phi(\omega) \rangle$. Bei Physikern und Ingenieuren ist auch üblich, stattdessen die Notation $\int T(t) \phi(t) dt$ zu verwenden. Sie ist für lokal-integrierbare Funktionen T , die man als reguläre Distributionen auffasst, sinnvoll. In Fällen, bei denen T eine singuläre Distribution ist, ist *das Integral nur eine formale Schreibweise*, die auf die Eigenschaften des Integrals als Linearform Bezug nimmt. Auch in *Mathematica* findet diese Notation Verwendung, wie wir sehen werden.

2.1 Fouriertransformationen absolut-integrierbarer Funktionen

2.1.1 Beispiele zur FT reellwertiger gerader und reellwertiger ungerader Funktionen, Symmetrie-Eigenschaften ihres Spektrums, Auswirkung einer Skalierung und einer Translation auf das Spektrum am Beispiel einer Rechteckfunktion

Beispiel 1. Wir betrachten die FT der Funktion $r[1,t]$ und der Skalierung $r[a,t]$ mit $a=1/2$.

Das Spektrum verbreitert sich, d.h. die "Bandbreite" wird umso größer je kleiner der Träger wird, und die Absolutwerte werden mit $1/|a|$ multipliziert. Die Gestalt des Spektrums bleibt *ähnlich* erhalten.


Eine quantitative Aussage über den Zusammenhang zwischen "Zeitdauer" und "Bandbreite" gibt die **Heisenbergsche Unschärferelation**, zu der ich später noch detaillierter komme. Dort wird "Zeitdauer" und "Bandbreite" näher erklärt

Da die Funktionen gerade reellwertige Funktionen sind, ist ihr Spektrum rein reell und ebenfalls gerade.

```
In[8]:= fourr1[ω_] = FourierTransform[r[1, t], t, ω, FourierParameters -> {1, -1}]
Plot[fourr1[ω], {ω, -8 π, 8 π}, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
```


Out[8]= $\frac{2 \sin[\omega]}{\omega}$

Out[9]=



```
In[10]:= fourr2[ω_] = FourierTransform[r[1/2, t], t, ω, FourierParameters → {1, -1}]
(* Spektrum reell und gerade *)
Plot[fourr2[ω], {ω, -8 π, 8 π}, PlotStyle → Directive[
  Blue, Thickness[0.006]], PlotRange → All]
```

Out[10]=
$$\frac{2 \operatorname{Sin}\left[\frac{\omega}{2}\right]}{\omega}$$




Beispiel 2. Nun die FT einer ungeraden reellwertigen Funktion. Ihr Spektrum ist rein imaginär und ungerade.

```


In[12]:= fungerade[t_] := r[1, t] Sign[t] (* ungerade reellwertige Funktion *)
Plot[fungerade[t], {t, -2, 2}, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
fourfungerade[ω_] =
  FourierTransform[fungerade[t], t, ω, FourierParameters -> {1, -1}]
(* Spektrum rein imaginär und ungerade*)
Plot[Im[fourfungerade[ω]], {ω, -8 π, 8 π}, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]

```



Out[14]=

$$\frac{i e^{-i \omega} (-1 + e^{i \omega})^2}{\omega}$$





Beispiel 3. Spektrum einer **Translation** der Rechteckfunktion zu einer **kausalen** Funktion f , d.h. mit Träger in $[0, \infty[$. Die Translation hat dasselbe *Amplitudenspektrum* $|F f| = |F r[1, \cdot]|$ wie die FT von $r[1, t]$, ihr Spektrum wird aber mit $e^{-i\omega}$ multipliziert als Folge der Verschiebung von $r[1, t]$ um 1 nach rechts, d.h. das Phasenspektrum $\arg(F f)$ wird komplett verändert.

Da die Funktion kausal ist besteht auch ein Zusammenhang zwischen Real- und Imaginärteil der FT von f (vgl. [1], S. 227). Dieser Sachverhalt spielt eine große Rolle beim Entwurf von Schaltungen, sagt er doch, dass beim Entwurf kausaler linearer Filter Real- und Imaginärteil des Frequenzgangs nicht unabhängig voneinander gewählt werden können.



```
In[16]:= fourrtrans[ω_] = FourierTransform[r[1, t - 1], t, ω, FourierParameters → {1, -1}]
(* FT der um 1 nach rechts verschobenen Rechteckfunktion r[1,t] *)
```

$$\text{Out[16]= } \frac{-2 i \sin[\omega]^2 + \sin[2 \omega]}{\omega}$$

```
In[17]:= Plot[r[1, t - 1], {t, -1, 3}, PlotStyle → Directive[
  Blue, Thickness[0.006]], PlotRange → All]
(* zeigt Treanslation von r[1,.] um 1 nach rechts *)
Plot[Abs[fourrtrans[ω]], {ω, -8 π, 8 π}, PlotStyle → Directive[
  Blue, Thickness[0.006]], PlotRange → All]
(* zeigt gleiches Amplitudenspektrum wie F r[1,.] *)
```



```
In[19]:= Plot[Arg[fourr1[ $\omega$ ]], { $\omega$ , -8  $\pi$ , 8  $\pi$ }, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
(* Phasenspektrum der FT von r[1,t] *)
Plot[Arg[fourrtrans[ $\omega$ ]], { $\omega$ , -8  $\pi$ , 8  $\pi$ }, PlotStyle -> Directive[
  Blue, Thickness[0.006]], PlotRange -> All]
(* Phasenspektrum der FT von r[1,t-1] *)
```



```
In[21]:= Simplify[fourrtrans[ $\omega$ ] / fourr1[ $\omega$ ]] (* Spektrum von r[1,t]
  wird wegen Translation zu r[1,t-1] mit  $e^{-i\omega}$  multipliziert,
  hier ist der Faktor  $e^{-i\omega}$  als Quotient zu sehen *)
```

Out[21]= $\text{Cos}[\omega] - i \text{Sin}[\omega]$

Da $f(t)=r[1,t-1]$ kausal ist, d.h. Träger in $[0,\infty[$ hat, hängen Realteil R und Imaginärteil X seiner FT zusammen. Es gilt für $t>0$

$$f(t) = \frac{2}{\pi} \int_0^{\infty} R(\omega) \cos(\omega t) d\omega - \frac{2}{\pi} \int_0^{\infty} X(\omega) \sin(\omega t) d\omega \quad (\text{für } t>0).$$

Nachfolgend diese Beziehung am Beispiel mit *Mathematica*:

```
In[22]:= f1[t_] =  $\frac{2}{\pi}$  Integrate[Re[fourrtrans[ $\omega$ ]] Cos[ $\omega$  t], { $\omega$ , 0,  $\infty$ }]
```


Out[22]= ConditionalExpression[$\frac{1}{2}$ (Sign[2 - t] + Sign[2 + t]), t ∈ Reals]

```
In[23]:= f2[t_] = - $\frac{2}{\pi}$  Integrate[Im[fourrtrans[ $\omega$ ]] Sin[ $\omega$  t], { $\omega$ , 0,  $\infty$ }]
```


```
Out[23]= ConditionalExpression[ $\frac{1}{2}$  (Sign[2 - t] + 2 Sign[t] - Sign[2 + t]), t  $\in$  Reals]
```

Dass f1 und f2 für $t > 0$ gleich sind, sehen Sie an den beiden folgenden Grafiken:

```
In[24]:= Plot[f1[t], {t, 0, 3}, PlotStyle  $\rightarrow$  Directive[
  Blue, Thickness[0.006]], PlotRange  $\rightarrow$  All]
```



```
In[25]:= Plot[f2[t], {t, 0, 3}, PlotStyle  $\rightarrow$  Directive[
  Blue, Thickness[0.006]], PlotRange  $\rightarrow$  All]
```



2.1.2 Spektrum bei Amplitudenmodulation

Beispiel 1. Wir demonstrieren zunächst, dass die **Gaussfunktion** $e^{-t^2/2}$ eine

Eigenfunktion der FT zum Eigenwert $\sqrt{2\pi}$ ist (mit unserer Wahl FourierParameters $\rightarrow \{1, -1\}$) und sehen uns die FT der Modulation $e^{i\omega_0 t} e^{-t^2/2}$ an. Das Spektrum der Gaussfunktion wird um ω_0 nach rechts verschoben. Wir plotten mit $\omega_0 = 5$.

```
In[26]:= fourgauss[ $\omega$ _] = FourierTransform[gauss[t], t,  $\omega$ , FourierParameters  $\rightarrow$  {1, -1}]
  (* zeigt, dass gauss[t] eine Eigenfunktion der FT zum EW  $\sqrt{2\pi}$  ist *)
```

```
Out[26]=  $e^{-\frac{\omega^2}{2}} \sqrt{2\pi}$ 
```

In[27]:= $e^{-\frac{\omega^2}{2}} \sqrt{2\pi}$

Out[27]:= $e^{-\frac{\omega^2}{2}} \sqrt{2\pi}$

In[28]:= `fourgaussmod1[$\omega_$] =`

`FourierTransform[Exp[5 i t] gauss[t], t, ω , FourierParameters \rightarrow {1, -1}]`

`(* zeigt, dass gauss[ω] um 5 nach rechts verschoben wird *)`

Out[28]:= $e^{-\frac{1}{2}(-5+\omega)^2} \sqrt{2\pi}$


Beispiel 2. Nun **Amplitudenmodulation eines cos(5t)**. Die Funktion $\cos(2\pi 5 t) e^{-t^2/2}$ ist eine 5 HZ-Cos-Schwingung zwischen den Hüllkurven $\pm e^{-t^2/2}$. Das Spektrum der modulierten Schwingung hat nun zwei Seitenbänder, die aus dem nach links und nach rechts verschobenen Spektrum von gauss[t] bei halbierten Amplitudenwerten bestehen. Die Maximalwerte im (reellen) Spektrum sind dann $\sqrt{2\pi}/2 \approx 1.25$ an den Stellen $\pm 10\pi$.

In[29]:= `Plot[{Cos[2 Pi 5 t] Exp[-t^2/2], Exp[-t^2/2], -Exp[-t^2/2]},`

`{t, -2, 2}, PlotStyle \rightarrow Directive[`

`Blue, Thickness[0.006]], PlotRange \rightarrow All]`

`(* die modulierte Schwingung und ihre Hüllkurven *)`




In[30]:= `fourcosmod[$\omega_$] =`

`ComplexExpand[FourierTransform[(Exp[i 2 Pi 5 t] + Exp[-i 2 Pi 5 t]) / 2 gauss[t],`

`t, ω , FourierParameters \rightarrow {1, -1}]]`

Out[30]= $\frac{\sqrt{\frac{\pi}{2}}}{\sqrt{e^{(-10\pi+\omega)^2}}} + \frac{\sqrt{\frac{\pi}{2}}}{\sqrt{e^{(10\pi+\omega)^2}}}$

```
In[31]:= Plot[fourcosmod[ $\omega$ ], { $\omega$ , -12  $\pi$ , 12  $\pi$ }, PlotRange  $\rightarrow$  All, PlotStyle  $\rightarrow$  Directive[
Blue, Thickness[0.006]], PlotRange  $\rightarrow$  All]
```



2.1.3 Mehrfache Fouriertransformation, Eigenwerte der FT, weitere Beispiele von Eigenfunktionen der FT, Inverse Fouriertransformation

Beispiel 1. Wir wenden die FT 4-mal auf die Dreieckfunktion $f[t]=d[1,t]$ wie oben definiert an. Die FT von $d[1,t]$ ist der **Fejerkern** $4 \sin^2(\omega/2)/\omega^2$. Das Beispiel demonstriert, dass für die Inverse Fouriertransformation F^{-1} gilt: $F^{-1}(Ff) = (4\pi^2)^{-1} F^3(Ff)$ ist.

```
In[32]:= fourd1[ $\omega_$ ] = FourierTransform[d[1, t], t,  $\omega$ , FourierParameters  $\rightarrow$  {1, -1}]
(* die FT von d[1,t] und der Test, dass dies der genannte Fejerkern ist *)
```

Out[32]=
$$\frac{2 - 2 \cos[\omega]}{\omega^2}$$

```
In[33]:= Simplify[fourd1[ $\omega$ ] - 4 sin[ $\omega/2$ ]2/ $\omega^2$ ]
```

Out[33]= 0



```
In[34]:= fourd2[ω_] = FourierTransform[fourd1[t], t, ω, FourierParameters → {1, -1}]
(* 2. FT *)
fourd3[ω_] = FourierTransform[fourd2[t], t, ω, FourierParameters → {1, -1}]
(* 3. FT *)
fourd4[ω_] = FourierTransform[fourd3[t], t, ω, FourierParameters → {1, -1}] / (4 π²)
(* 4. FT / (4π²) führt zurück auf d[1,t] *)
```

Out[34]= $\pi \text{Sign}[1 - \omega] - \pi \omega \text{Sign}[1 - \omega] - 2 \pi \omega \text{Sign}[\omega] + \pi \text{Sign}[1 + \omega] + \pi \omega \text{Sign}[1 + \omega]$

Out[35]= $-\frac{4 \pi (-1 + \text{Cos}[\omega])}{\omega^2}$

Out[36]= $-\frac{1}{\pi} \left(-\frac{1}{2} \pi \text{Sign}[1 - \omega] + \frac{1}{2} \pi \omega \text{Sign}[1 - \omega] + \pi \omega \text{Sign}[\omega] - \frac{1}{2} \pi \text{Sign}[1 + \omega] - \frac{1}{2} \pi \omega \text{Sign}[1 + \omega] \right)$

```
In[37]:= Plot[fourd1[ω], {ω, -4 π, 4 π}, PlotRange → All, PlotStyle → Directive[
Blue, Thickness[0.006]]]
Plot[fourd4[ω], {ω, -2, 2}, PlotRange → All, PlotStyle → Directive[
Blue, Thickness[0.006]]]
```



Beispiel 2. (Beispiele zu Eigenfunktionen der Fouriertransformation)
mit der Gaussfunktion hatten wir oben bereits eine Eigenfunktion der FT gesehen. Die FT hat (in unserer Normierung) nur die 4 Eigenwerte $\pm\sqrt{2 \pi}$ und $\pm i \sqrt{2 \pi}$. In diesem Beispiel sehen wir, dass z.B. auch

a) die Hermite-Funktionen $h_n(t) = (-1)^n e^{t^2/2} \frac{d^n}{dt^n} e^{-t^2}$

oder

b) die Funktion $1 / \cosh(\sqrt{\pi/2} t)$ Eigenfunktionen der FT sind.

(Unter den Distributionen aus S' ist z.B. auch $1 / \sqrt{|t|}$ eine "Eigendistribution" wie wir später noch sehen werden; zu den Beweisen siehe [1]).

Zu a)

Mathematica stellt die **Hermite-Polynome** als **HermiteH[n,t]** schon zur Verfügung, die **Hermite-Funktionen** sind dann $h[n,t]=e^{-t^2/2}$ **HermiteH[n, t]**.

Nun zur FT von $h[n,t]$, konkret etwa für $n=4$: Wir dividieren die FT von $h[4,t]$ als Funktion von ω durch $(-i)^4 \sqrt{2\pi} h[4,\omega]$ und erhalten 1, d.h. $(-i)^4 \sqrt{2\pi}$ ist FT-Eigenwert für $h[4,t]$, allgemein $(-i)^n \sqrt{2\pi}$ FT-Eigenwert für $h[n,t]$:

```
In[39]:= h[n_, t_] = e^{-t^2/2} HermiteH[n, t]
```

```
Out[39]= e^{-t^2/2} HermiteH[n, t]
```

```
In[40]:= fourhn[\omega_] =
```

```
Simplify[FourierTransform[h[4, t], t, \omega, FourierParameters -> {1, -1}] /
  ((-i)^4 \sqrt{2\pi} h[4, \omega])]
```

```
Out[40]= 1
```

Die Hermite - Funktionen bilden, geeignet in $L^2(\mathbb{R})$ normiert, ein vollständiges Orthonormalsystem von Eigenfunktionen für den Hamilton-Operator des harmonischen Oszillators in der Quantenmechanik. Die zugehörigen Eigenwerte, in denen das Plancksche Wirkungsquantum vorkommt, sind die diskreten Energieniveaus des harmonischen Oszillators. (man vgl. Literatur zur Quantenmechanik, z.B. die unten genannte Referenz [2]).

Zu b)

Mathematica notiert $1/\cosh[x]$ als $\text{Sech}[x]$. Nachfolgend sehen wir, dass

$1 / \cosh[\sqrt{\pi/2} t]$ FT-Eigenfunktion zum Eigenwert $\sqrt{2\pi}$ ist:

```
In[41]:= FourierTransform[1 / Cosh[\sqrt{\pi/2} t], t, \omega, FourierParameters -> {1, -1}]
```

```
Out[41]= \sqrt{2\pi} Sech[\sqrt{\frac{\pi}{2}} \omega]
```

In[42]:=

2.1.4 Fouriertransformation von Ableitungen Zusammenhänge zwischen Spektrum und Differenzierbarkeitseigenschaften transformierter Funktionen, Einfluss von Störungen auf ein Spektrum

Für $f \in L^1(\mathbb{R})$ ist $F(f)$ stetig und beschränkt mit $F(f)(\omega) \rightarrow 0$ für $|\omega| \rightarrow \infty$ (Riemann-Lebesgue-Lemma).
Für differenzierbare Funktionen $f \in L^1(\mathbb{R})$ mit $f' \in L^1(\mathbb{R})$ ist $F(f')(\omega) = i\omega F(f)$, F die FT.
Ist $g(t) = t^k f(t)$ integrierbar, dann ist $F(g)$ k -mal stetig differenzierbar.
Anders gesagt: *Je glatter f ist, desto schneller fällt f für $|\omega| \rightarrow \infty$ ab. Umgekehrt können auch kleine nicht-differenzierbare Störungen das Spektrum einer Funktion massiv verändern.* Wir testen als Beispiel mit der Gaußfunktion.

Beispiel 1. FT der Ableitung der Gaußfunktion

In[43]:= `FourierTransform[D[gauss[t], t], t, \omega, FourierParameters -> {1, -1}]`

Out[43]= $i e^{-\frac{\omega^2}{2}} \sqrt{2\pi} \omega$

Beispiel 2. (Verrauschung des Spektrums) Amplitudenspektrum der Funktion $f[t] = e^{-|t|}$ (blau) und Spektrum der Funktion $f[t] + \text{stoerung}[t]$ (rot), wobei wir als **Störung** ein paar Rechteckimpulse $r_k[t]$ mit $\int r_k[t] dt = 1$ von nur kurzer Zeitdauer aufsetzen.
Das rote Amplitudenspektrum macht deutlich, wie sehr z.B. Interrupts (kurze Spannungsänderungen) in einem elektrotechnischen Gerät das Spektrum eines dort anliegenden Signals mit beträchtlichen Amplituden stören können.

```


In[44]:= fourf[ω_] = FourierTransform[Exp[-Abs[t]], t, ω, FourierParameters → {1, -1}]
(* Spektrum von f *)
fourstoer[ω_] =
  FourierTransform[103 r[10-3, t - 1] + 103 r[10-3, t - 3] + 103 r[10-3, t - 5],
  t, ω, FourierParameters → {1, -1}]
(* Spektrum des gestöten Signals *)
amp1 :=
  Plot[Abs[fourf[ω]], {ω, -20, 20}, PlotRange → {0, 10}, PlotStyle → Directive[
    Blue, Thickness[0.006]]]
amp2 := Plot[Abs[fourf[ω] + fourstoer[ω]],
  {ω, -20, 20}, PlotRange → {0, 10}, PlotStyle → Directive[
    Red, Thickness[0.006]]]
Show[{amp1, amp2}]

```

$$\text{Out[44]= } \frac{2}{1 + \omega^2}$$

$$\text{Out[45]= } -\frac{1000 i e^{-\frac{999 i \omega}{1000}}}{\omega} + \frac{1000 i e^{-\frac{1001 i \omega}{1000}}}{\omega} - \frac{1000 i e^{-\frac{2999 i \omega}{1000}}}{\omega} +$$

$$\frac{1000 i e^{-\frac{3001 i \omega}{1000}}}{\omega} - \frac{1000 i e^{-\frac{4999 i \omega}{1000}}}{\omega} + \frac{1000 i e^{-\frac{5001 i \omega}{1000}}}{\omega}$$



Praxisrelevanz: Als Beispiel sei erzählt, dass etwa der Anschlag auf dem E-Piano bei mir zuhause regelmäßig die Hausklingel zum Läuten gebracht hat, bis ich ein Filter an den Klingeldraht angebracht habe. Grund: Anregende Amplituden bei der Klingelanregungsfrequenz durch die Störung im Stromnetz durch die Anschlagimpulse vom E-Piano her. Ganz ähnliche folgenreichere Effekte vom Materialaufbrauch bei Kupferleitungen bis hin zum Kabelbrand können viele Leistungsgeräte in den Stromnetzen von Firmen verursachen, wenn keine Gegenmaßnahmen ergriffen werden. Recherchieren Sie einmal dazu unter dem Stichwort **“intermittierende Frequenzen”**.

Rauschen: Auch Quantisierungen wie sie in der Bild- oder Audio-Datenverarbeitung bei JPEG oder MP3, digitalen Videos usw. durchgeführt werden, entsprechen pro Sample einer Verfälschung (Addition oder Subtraktion) mit einem kleinen Impuls. Als Folge solcher Verfälschungen (i.d.R. bei einer sehr großen Anzahl von Samples) wird das gesamte

Spektrum über die ganze Bandbreite des Kanals verfälscht, wie wir gerade gesehen haben. Der Effekt wird als “**Quantisierungsrauschen**” bezeichnet. Unter diesem Stichwort finden Sie z.B. bei Wikipedia Hörbeispiele mit unterschiedlichen Quantisierungen (24 Bit, 12 Bit, .. 2 Bit) einer Audio-Aufnahme. Dabei bekommen Sie nicht wie oben einen bildklichen, sondern akustischen Eindruck des Effekts.

Andererseits lässt sich “Rauschen” auch verwenden, um dem Spektrum z.B. eine Nachricht aufzuprägen, die geheim bleiben soll und nur vom Empfänger mit Kenntnis der Methode wieder entdeckt werden kann. Solche Methoden wurden in der militärischen Nachrichtenübertragung schon seit 1950 genutzt und finden heute auch Verwendung beim Thema “Digitale Wasserzeichen” im Digital Rights Management.

2.1.5 Parsevalgleichung, Heisenbergsche Unschärferelation

Eine weitere grundlegende Beziehung zwischen Funktionen und ihren Fouriertransformierten ist die **Parsevalgleichung**, die mit unserer Normierung der FT wie folgt lautet:

$$\int |f(t)|^2 dt = \frac{1}{2\pi} \int |F(f)(\omega)|^2 d\omega.$$

Sie gilt für alle $f \in L^2(\mathbb{R})$. Dabei ist die Fouriertransformation eine stetige lineare und bijektive Fortsetzung der Fouriertransformation vom Schwartzschen Raum S der schnell fallenden Funktionen auf ganz $L^2(\mathbb{R})$. Zu Details vgl. [1],Kap. 9.

Wir betrachten die Parsevalgleichung zunächst am Beispiel der Dreieckfunktion $f(t) = d[1, t]$ (vgl. ganz oben) und danach für f' . Dies führt uns zur **Heisenbergschen Unschärferelation**.

Beispiel 1. Parsevalgleichung für die Dreieckfunktion $d[1, t]$. Wir rechnen beide Seiten der Parsevalgleichung mit *Mathematica* aus:

```
In[49]:= Integrate[Abs[d[1, t]]^2, {t, -1, 1}]
Integrate[ $\frac{1}{2\pi}$ 
Abs[FourierTransform[d[1, t], t, \omega, FourierParameters -> {1, -1}]]^2, {\omega, -\infty, \infty}]
Out[49]=  $\frac{2}{3}$ 
Out[50]=  $\frac{2}{3}$ 
```

Beispiel 2. Parsevalgleichung für die Ableitung der Dreieckfunktion $f[t]=d[1,t]$. Wir rechnen beide Seiten der Parsevalgleichung mit *Mathematica* aus. Danach vergleichen wir mit dem Integral über $|\omega F(f)(\omega)|^2 / (2\pi)$. Feststellung: Alle Integrale ergeben denselben Wert. Dies gilt generell für alle $f \in L^2(\mathbb{R})$, sobald die Integrale existieren.

```

In[51]:= Integrate[Abs[D[d[1, t], t]]^2, {t, -1, 1}]
Integrate[
  1
  Abs[FourierTransform[D[d[1, t], t], t, ω, FourierParameters → {1, -1}]]^2,
  2 π
  {ω, -∞, ∞}]
Integrate[
  1
  Abs[ω FourierTransform[d[1, t], t, ω, FourierParameters → {1, -1}]]^2,
  2 π
  {ω, -∞, ∞}]
Out[51]= 2
Out[52]= 2
Out[53]= 2

```

Nun zur Heisenbergschen Unschärferelation für das Zeitdauer-Bandbreite-Produkt

Die **effektive Zeitdauer** $D_t(f)$ für eine L^2 -Funktion $f \neq 0$ ist definiert durch (vgl. [1], Abschnitt 11.2)

$$D_t(f) = \Delta_a(f),$$

$$\text{wobei } \Delta_a^2(f) = \frac{\int (t-a)^2 |f(t)|^2 dt}{\int |f(t)|^2 dt} \text{ und } a = \frac{\int t |f(t)|^2 dt}{\int |f(t)|^2 dt} \text{ sind.}$$

Analog ist die effektive Bandbreite $D_\omega(f)$ von f mit der FT von f definiert durch

$$D_\omega(f) = \Delta_b(Ff),$$

$\Delta_b(Ff)$ analog zu $\Delta_a(f)$ durch Ersetzen von a durch b und von f durch Ff .

Interpretation:

Für eine Massendichte $|f|^2$ wäre a der Schwerpunkt und $\Delta_a^2(f)$ ein Trägheitsmoment, für eine Wahrscheinlichkeitsdichte $|f|^2$ wäre a der Erwartungswert und $\Delta_a^2(f)$ die Varianz der Wahrscheinlichkeitsverteilung.

Effektive Zeitdauer und Bandbreite geben ein Maß dafür, wie sehr f bzw. Ff um a bzw. b herum "konzentriert sind". **Die Heisenbergsche Unschärferelation für das Zeitdauer-Bandbreite-Produkt lautet:**

Für quadratisch-integrierbare Signale $f \neq 0$ gilt für beliebige $a, b \in \mathbb{R}$ gilt

$$\Delta_a(f) \Delta_b(Ff) \geq \frac{1}{2}$$

Beispiel 3. Wir betrachten zunächst die Dreieckfunktion $f[t]=d[1,t]$ wie zuletzt. Sie hat als "Erwartungswert"/"Schwerpunkt" $a=0$, für ihre Fouriertransformierte, den **Fejerkern**, gilt analog $b=0$. Wir berechnen Zeitdauer, Bandbreite und das Produkt davon:

```
In[54]:= Integrate[t Abs[d[1, t]]^2, {t, -1, 1}] (* zeigt a=0 *)
Integrate[ω Abs[FourierTransform[d[1, t], t, ω, FourierParameters → {1, -1}]]^2,
{ω, -∞, ∞}] (* zeigt b=0 *)
```

Out[54]= 0

Out[55]= 0

```
In[56]:= dtf = Sqrt[Integrate[t^2 Abs[d[1, t]]^2, {t, -1, 1}] /
Integrate[Abs[d[1, t]]^2, {t, -1, 1}]] (* effektive Zeitdauer *)
dwf = Sqrt[Integrate[
ω^2 Abs[FourierTransform[d[1, t], t, ω, FourierParameters → {1, -1}]]^2,
{ω, -∞, ∞}] / (2 π Integrate[Abs[d[1, t]]^2, {t, -1, 1}])]
(* effektive Bandbreite von d[1,t] *)
```

Out[56]= $\frac{1}{\sqrt{10}}$

Out[57]= $\sqrt{3}$

Ergebnis: Das Zeitdauer-Bandbreite-Produkt ist größer als 1/2:

```
In[58]:= produkt = N[dtf dwf]
```

Out[58]= 0.547723

Wir betrachten noch kurz eine Translation der obigen Dreieckfunktion. Dies ändert nicht effektive Zeitdauer und damit auch nicht das Zeitdauer-Bandbreite-Produkt:

```
In[59]:= normsqr = Integrate[Abs[d[1, t - 2]]^2, {t, 1, 3}]
(* Die quadrierte L^2-Norm von d[1,t-2] *)
```

Out[59]= $\frac{2}{3}$

```
In[60]:= a = Integrate[t Abs[d[1, t - 2]]^2, {t, 1, 3}] / normsqr
(* berechnet den "Schwerpunkt" von d[1,t-2], der natürlich 2 ist *)
b =
Integrate[ω Abs[FourierTransform[d[1, t - 2], t, ω, FourierParameters → {1, -1}]]^2,
{ω, -∞, ∞}] / (2 π normsqr)
(* Die Verschiebung ändert nicht den Schwerpunkt der FT von d[1,t],
da dadurch nur das Phasenspektrum verändert wird *)
```

Out[60]= 2

Out[61]= 0

Hier noch die berechnete effektive Zeitdauer von $d[1,t-2]$, welche die gleiche ist wie vorher. Auch die Bandbreite bleibt gleich und damit auch das Produkt aus beiden Größen.

```
In[62]:= dtf2 = Sqrt[Integrate[(t - a)^2 Abs[d[1, t - 2]]^2, {t, 0, 3}]/normsqr]
(* effektive Zeitdauer von d[1,t-2] *)
```

```
N[dtf2]
```

```
Out[62]= 1/√10
```

```
Out[63]= 0.547723
```

Anschauliche Erklärung des Sachverhalts: Eine *Kompression* etwa von $d[1,t]$ zu $d[1,5t]$ hat *wachsende Steigungen* zur Folge. Nach der Parseval-Gleichung und der Regel für die FT von Ableitungen gilt

$$2\pi \int |f'(t)|^2 dt = \int |F(f')(\omega)|^2 d\omega = \int \omega^2 |F(f)(\omega)|^2 d\omega$$

Daher sieht man unmittelbar, dass mit $|f'|$ *wachsende Steigungen die Bandbreite vergrößern*. Eine quantitative Beschreibung der Relation zwischen Zeitdauer (Kompression im Zeitbereich) und Bandbreite (Kompression im Frequenzbereich) gibt die Heisenbergsche Unschärferelation.

Sie setzt eine Schranke, die es z.B. in der Nachrichtentechnik nicht erlaubt, bei gegebener Bandbreite eines Übertragungskanal eine nach der Unschärferelation beschränkte Übertragungskapazität zu überschreiten. Große Übertragungsgeschwindigkeiten brauchen daher entsprechend große Bandbreiten. Je mehr Nutzer sich etwa eine DSL- oder LTE-Übertragungseinrichtung mit vom Anbieter fest vorgegebener Bandbreite teilen müssen, desto geringer wird zwangsläufig die Übertragungsrate pro Nutzer.

Die Unschärferelation gilt auch für die Genauigkeit von Ort und Impuls eines quantenmechanischen Systems und hat auch hier ihre Ursache in Eigenschaften der Fouriertransformation. **Sie ist also insbesondere nicht durch Mess-Ungenauigkeiten verursacht, sondern ist eine grundlegende Eigenschaft von Funktionen und ihren Fouriertransformierten.**

Das Zeitdauer-Bandbreite-Produkt wird genau 1/2, wenn f bis auf die Phase eine Gaußfunktion ist, d.h. wenn $f(t) = c e^{iat} e^{-(t-m)^2/(4\sigma^2)}$ mit beliebigen Konstanten a,m, $c \neq 0$ und $\sigma > 0$ ist.

2.1.6 Zum Gibbsphänomen bei der Fouriertransformation

Das Gibbsche Phänomen das von Fourierreihen her bekannt ist, gilt auch bei der Fouriertransformation. Wir sehen auch hier bei der Näherung einer ansonsten stetigen integrierbaren Funktion f in der Umgebung von Sprungstellen ein Überschwingen um ca. 9% der Sprunghöhe, wenn wir statt der Fourierschen Umkehrformel zur Rekonstruktion von f


aus dem Spektrum die **Näherung**

$$\frac{1}{2\pi} \int_{-\Omega}^{+\Omega} F(f)(\omega) e^{i\omega t} d\omega$$

verwenden.

Beispiel. Wir betrachten eine Approximation für die Rechteckfunktion $r_2[t]=10 r[1,t]$ mit $\Omega=20$ und sehen uns die entsprechende Grafik an. Die FT von r_2 hatten wir oben bis auf den Faktor 10 als $\text{fourr1}[t]$ schon berechnet (Die numerischen Integrationen mögen je nach Rechner etwas dauern, bei mir gerade ca. 1/2 Minute). Den ca. 9% Fehler in der Umgebung der Sprungstellen nach oben und nach unten kann man gut ablesen. Man kann das Überschwingen durch einen Faltungskern wie den Fejerkernel vermeiden (vgl. auch [1], Kap.9).

```
In[64]:= plot1 := Plot[10 r[1, t], {t, -3, 3}, PlotRange -> {-1, 12}, PlotStyle -> Directive[
  Blue, Thickness[0.006]]]
plot2 := Plot[1 / (2 π) NIntegrate[10 fourr1[ω] Exp[i ω t], {ω, -10, 10}],
  {t, -3, 3}, PlotRange -> {-1, 12}, PlotStyle -> Directive[
  Red, Thickness[0.006]]]
Show[{plot1, plot2}]
```



Bevor wir auf einen der wichtigsten Sachverhalte, nämlich den **Faltungssatz für die Fouriertransformation** zu sprechen kommen, wollen wir noch einige Beispiele für **Fouriertransformationen temperierter Distributionen** ansehen. **Erst im Zusammenwirken mit der Distributionentheorie entfaltet die Fouriertransformation ihre ganze Kraft.** Dieses Zusammenwirken hat seit der Entwicklung dieser Theorie ab etwa 1950 einen historisch gut nachvollziehbaren Schub in der Technik-Entwicklung hervorgebracht und hält bis heute an.

2.2 Fouriertransformationen von Distributionen aus S'

Bei allem Nutzen, den die klassische Fourier-Analyse auch schon vor der Entwicklung der Distributionentheorie hatte, gewinnt sie doch erst an Kraft, sobald auch für typische Funktionen in Anwendungsdisziplinen wie $\sin(t)$, eine Sprungfunktion, Polynome, rationale Funktionen oder dann auch Impulsfolgen als typische singuläre Distributionen die Fouriertransformation eingeführt werden kann. Alle genannten Funktionen haben keine Fouriertransformierte, da sie nicht absolut-integrierbar sind und auch nicht zu $L^2(\mathbb{R})$ gehören. Sie gehören aber alle zum Raum S' der temperierten Distributionen. Für Einzelheiten zu den Räumen S der schnell fallenden Funktionen und S' sei auf [1] verwiesen. Die Anwendung einer Distribution T auf eine Testfunktion ϕ notiere ich wie in vorangehenden Notebooks durch $\langle T, \phi \rangle$.

Nachfolgend verwenden wir die Fouriertransformation auf dem Raum S' , definiert für Distributionen $T \in S'$ und Testfunktionen $\phi \in S \subset L^1(\mathbb{R})$ durch

$$\langle F(T), \phi \rangle = \langle T, F(\phi) \rangle.$$

Die Fouriertransformation ist dann auf ganz S' definiert und dort stetig und umkehrbar.

Auch *Mathematica* kann Distributionen aus S' - sofern sie nicht zu kompliziert sind - der Fouriertransformation unterziehen, d.h. kennt den Distributionenkalkül. Wir betrachten einige Beispiele. Wie man diese Fouriertransformationen herleitet können sie in [1] finden.

2.2.1 Beispiele für Fouriertransformationen temperierter Distributionen

Beispiel 1. Fouriertransformation der DiracDistribution $\delta(t - t_0)$ ist die Funktion $e^{-i\omega t_0}$, die der verallgemeinerten Ableitung ist $i\omega e^{-i\omega t_0}$. Im Beispiel wählen wir $t_0 = 3$. Damit wird schon klar, dass die FT einer Impulsfolge $\sum_{k=-\infty}^{+\infty} c_k \delta(t-ka)$ in S' die verallgemeinerte Fourierreihe $\sum_{k=-\infty}^{+\infty} c_k e^{-ik\omega a}$ ergibt.

```
In[67]:= fourdelta[ω_] =
  FourierTransform[DiracDelta[t - 3], t, ω, FourierParameters -> {1, -1}]
fourddelta[ω_] = FourierTransform[D[DiracDelta[t - 3], t],
  t, ω, FourierParameters -> {1, -1}]
```


```
Out[67]= e-3 i ω
```

```
Out[68]= i e-3 i ω ω
```

Beispiel 2. Fouriertransformierte der Funktion $f(t) = \sin(t)/t$. Die Funktion f ist nicht absolut-integrierbar, gehört aber zu $L^2(\mathbb{R}) \subset S'$ und ist daher Fourier-transformierbar. Ihre FT ist eine Rechteckfunktion. Die Fouriertransformierte beschreibt in der linearen Systemtheorie ein ideales Tiefpassfilter (das sog. *Küpfmüller-Tiefpassfilter*). Die Funktion f ist dann die (theoretische) Impulsantwort des Filters. Sie zeigt, dass ein ideales Tiefpassfilter nicht kausal und damit in der Praxis nicht realisierbar ist.

```
In[69]= foursinc[ω_] = FourierTransform[Sin[t] / t, t, ω, FourierParameters → {1, -1}]
Plot[foursinc[s], {s, -2, 2}, PlotRange → {-1, 4}, PlotStyle → Directive[
  Blue, Thickness[0.006]]]
```

$$\text{Out[69]= } \frac{1}{2} \pi \text{Sign}[1 - \omega] + \frac{1}{2} \pi \text{Sign}[1 + \omega]$$



Ein Sachverhalt, den man an diesem Beispiel demonstriert sieht, ist eine Bemerkung wert: Die FT von f hat hier kompakten Träger, das Zeitsignal f ist unendlich oft differenzierbar und verschwindet auf keinem Intervall. Das gilt ganz allgemein: *Bandbegrenzte Signale sind immer unendlich oft differenzierbare Funktionen, die durch ihre Taylorreihen dargestellt werden, nach dem Identitätssatz für Potenzreihen also auf keinem Intervall verschwinden können.*

Beispiel 3. Die Fouriertransformierte der Heaviside-Funktion

```
In[71]= fourheavi[ω_] =
  FourierTransform[HeavisideTheta[t], t, ω, FourierParameters → {1, -1}]
```

$$\text{Out[71]= } -\frac{i}{\omega} + \pi \text{DiracDelta}[\omega]$$

Beispiel 4. Fouriertransformation eines Polynoms, als Beispiel $t^3 + 2t^2 + 3t + 5$

Allgemein gilt: $F\left(\sum_{k=0}^n a_k t^k\right) = 2\pi \sum_{k=0}^n i^k a_k \delta^{(k)}$

```
In[72]= fourpoly[ω_] = FourierTransform[t^3 + 2 t^2 + 3 t + 5, t, ω, FourierParameters → {1, -1}]
```

$$\text{Out[72]= } 10 \pi \text{DiracDelta}[\omega] + 6 i \pi \text{DiracDelta}'[\omega] - 4 \pi \text{DiracDelta}''[\omega] - 2 i \pi \text{DiracDelta}^{(3)}[\omega]$$

Beispiel 5. Fouriertransformation trigonometrischer Funktionen

```
In[73]:= foursina[ω_] = FourierTransform[Sin[A t], t, ω, FourierParameters → {1, -1}]
fourcosa[ω_] = FourierTransform[Cos[A t], t, ω, FourierParameters → {1, -1}]
fourexpa[ω_] = FourierTransform[Exp[i A t], t, ω, FourierParameters → {1, -1}]
```

```
Out[73]= -i π DiracDelta[A - ω] + i π DiracDelta[A + ω]
```

```
Out[74]= π DiracDelta[A - ω] + π DiracDelta[A + ω]
```

```
Out[75]= 2 π DiracDelta[A - ω]
```

Beispiel 6. Fouriertransformation trigonometrischer Funktionen auf der Halbachse t > 0

```
In[76]:= foursina[ω_] =
  FourierTransform[Sin[A t] HeavisideTheta[t], t, ω, FourierParameters → {1, -1}]
fourcosa[ω_] = FourierTransform[Cos[A t] HeavisideTheta[t],
  t, ω, FourierParameters → {1, -1}]
fourexpa[ω_] = FourierTransform[Exp[i A t] HeavisideTheta[t],
  t, ω, FourierParameters → {1, -1}]
```

```
Out[76]=  $\frac{1}{2(A - \omega)} + \frac{1}{2(A + \omega)} - \frac{1}{2} i \pi \text{DiracDelta}[-A + \omega] + \frac{1}{2} i \pi \text{DiracDelta}[A + \omega]$ 
```

```
Out[77]=  $\frac{i}{2(A - \omega)} - \frac{i}{2(A + \omega)} + \frac{1}{2} \pi \text{DiracDelta}[-A + \omega] + \frac{1}{2} \pi \text{DiracDelta}[A + \omega]$ 
```

```
Out[78]=  $\frac{i}{A - \omega} + \pi \text{DiracDelta}[-A + \omega]$ 
```

Beispiel 7. Inverse Fouriertransformation typischer Partialbrüche, wie sie in Frequenzgängen linearer Filter vorkommen: Wir wählen 3 typische mögliche Fälle für die Distribution

$$F(h)(\omega) = \frac{1}{(i\omega - (r + ib))^k}$$

Die Ergebnisse zeigen deutlich, welche Bedingungen die Pole einer rationalen Filter-Übertragungsfunktion vom Typ $1/(z - z_0)^k$ erfüllen müssen, damit die Übertragungsfunktion ein stabiles kausales Filter beschreibt. Dies ist genau dann der Fall, wenn die inverse FT von $F(h)$ - das ist dann die Impulsantwort des Filters - kausal und integrierbar ist (oder ggf. als Anteil noch einen Dirac-Impuls hat - das wäre der Fall mit $k=0$ in $F(h)$ oben). Mehr zu linearen Filtern finden Sie in einem nachfolgenden Notebook.

a) Fall einer komplexen Polstelle der Übertragungsfunktion mit negativem Realteil und Vielfachheit 1, etwa $r = -2 < 0$, $b=3$ (oder sonst irgendwie reell), $k=1$. Hier wäre eine wie oben angesprochene Impulsantwort h absolut-integrierbar. In den Übertragungsfunktionen mit reellen Koeffizienten käme auch der komplex-konjugierte Pol noch vor, so dass mit einem weiteren Partialbruch der gleichen Form in Summe eine reellwertige Impulsantwort resultieren würde.

```
In[79]:= invfourratio1[ω_] =
  InverseFourierTransform[1 / (i ω - (-2 + i 3)), ω, t, FourierParameters → {1, -1}]
Out[79]= e(-2+3 i) t HeavisideTheta[t]
```

b) Fall einer komplexen Polstelle der Übertragungsfunktion mit negativem Realteil und Vielfachheit $k=3$, dieselben r und b wie oben. Auch hier erhalten wir eine kausale integrierbare inverse FT.

```
In[80]:= invfourratio2[ω_] =
  InverseFourierTransform[1 / (i ω - (-2 + i 3))3, ω, t, FourierParameters → {1, -1}]
Out[80]=  $\frac{1}{2} e^{(-2+3 i) t} t^2 \text{HeavisideTheta}[t]$ 
```

c) In allen anderen Fällen mit $r \geq 0$ ergibt sich bei der inversen Fouriertransformation ein Funktion, die nicht mehr kausal ist, also nicht Bestandteil der Impulsantwort eines kausalen Filters sein kann. Wie betrachten als Beispiele noch die Fälle $r=0$ und $r=1$, b irgendwie reell, und $k=1$. Im ersten Fall $r=0$ hat die inverse FT ganz \mathbb{R} als Träger, im zweiten Fall $r>0$ den Träger $]-\infty, 0]$

```
In[81]:= invfourratio3[ω_] =
  InverseFourierTransform[1 / (i ω - i 3), ω, t, FourierParameters → {1, -1}]
invfourratio4[ω_] = InverseFourierTransform[
  1 / (i ω - (2 + i 3)), ω, t, FourierParameters → {1, -1}]
Out[81]=  $\frac{1}{2} e^{3 i t} \text{Sign}[t]$ 
Out[82]= -e(2+3 i) t HeavisideTheta[-t]
```

Beispiel 8. Fouriertransformation der verallgemeinerten Funktion $1 / \sqrt{|t|}$

Dieses Beispiel zeigt eine verallgemeinerte Eigenfunktion zum Eigenwert $\sqrt{2\pi}$

```
In[83]:= FourierTransform[1 / sqrt[Abs[t]], t, ω, FourierParameters → {1, -1}]
Out[83]=  $\frac{\sqrt{2\pi}}{\sqrt{\text{Abs}[\omega]}}$ 
```

2.2.2 Fouriertransformationen von Faltungen

Wir kommen nun zur vielleicht wichtigsten Eigenschaft der Fouriertransformation, nämlich zu ihrer Wirkung auf Faltungen. Unter geeigneten Bedingungen an die beteiligten Distributionen T und G gilt der **Faltungssatz für die Fouriertransformation F**

$$F(T * G) = F(T) \cdot F(G).$$

Da schon Faltungen von Distributionen nicht generell, sondern nur unter Zusatzbedingungen möglich sind, braucht es auch gewisse Bedingungen an die Distributionen T und G , damit der Faltungssatz gilt. Hinreichende Bedingungen, die für viele Anwendungen auch erfüllt sind, sind z.B.

- a) T und G sind absolut- oder quadratisch-integrierbar
- b) Eine der beiden Distributionen T oder G ist eine schnell fallende Funktion aus S'
- c) Eine der Distributionen T oder G hat kompakten Träger
- d) Eine der beiden Distributionen T oder G hat eine Fouriertransformierte, die unendlich oft differenzierbar und polynomial beschränkt (langsam wachsend) ist. Diese gehört dann zum Raum der Multiplikatoren in S' .

Faltungen bei kontinuierlichen Parametern

Wir betrachten zunächst Faltungen von Funktionen und Distributionen über dem Parameterraum \mathbb{R} . Im Blick haben wir dabei insbesondere Anwendungen wie die Systemtheorie für analoge lineare Filter.

Hierzu ein paar Beispiele.

Beispiel 1. Fouriertransformation der Faltung $f = (\text{HeavisideTheta} * \text{HeavisideTheta})$

Die FT der Faltung existiert, aber **der Faltungssatz gilt hier nicht**, da das Produkt $F(\text{HeavisideTheta})^2$ nicht existiert, denn in unserem Kalkül kann man keine Dirac-Distributionen mit einander multiplizieren.

```
In[84]:= fourheavi[ω_] =
  FourierTransform[HeavisideTheta[t], t, ω, FourierParameters -> {1, -1}]
(* die FT der Heaviside-Funktion *)
faltheavi[t_] = Convolve[HeavisideTheta[s], HeavisideTheta[s], s, t]
(* Die Faltung der Heaviside-Funktion mit sich selbst *)
fourhfalt[ω_] = FourierTransform[faltheavi[t], t, ω, FourierParameters -> {1, -1}]
(* Die FT dieser Faltung *)
```

$$\text{Out[84]} = -\frac{i}{\omega} + \pi \text{DiracDelta}[\omega]$$

$$\text{Out[85]} = t \text{HeavisideTheta}[t]$$

$$\text{Out[86]} = -\frac{1}{\omega^2} + i \pi \text{DiracDelta}'[\omega]$$

Der Versuch, das Produkt der FT von f zu bilden, liefert mit *Mathematica* zwar ein "formales Ergebnis", aber auch *Mathematica* kann dann nicht mehr zu f zurück transformieren, der Kalkül greift hier nicht mehr.

```
In[87]:= test[ω_] = fourheavi[ω] fourheavi[ω]
InverseFourierTransform[test[ω], ω, t, FourierParameters → {1, -1}]
```

```
Out[87]=  $\left(-\frac{i}{\omega} + \pi \text{DiracDelta}[\omega]\right)^2$ 
```

```
Out[88]= InverseFourierTransform[ $\left(-\frac{i}{\omega} + \pi \text{DiracDelta}[\omega]\right)^2$ , ω, t, FourierParameters → {1, -1}]
```

Beispiel 2. In Anwendungsbeispielen aus der Linearen Systemtheorie und Signalverarbeitung, die ich in nachfolgenden Notebooks noch vorstellen will, sind zeitinvariante Systeme vielfach durch lineare Differentialgleichungen mit konstanten Koeffizienten der Form $P(D)u=f$ beschrieben. Von praktischem Interesse sind dabei kausale und stabile Systeme.

In einem vorangehenden Notebook über Anwendungen der Distributionentheorie habe ich schon gezeigt (vgl. auch [1], Kap.8), dass es zu solchen Gleichungen immer genau eine kausale Grundlösung g gibt.

Ist das System asymptotisch stabil, d.h. hat das charakteristische Polynom P der DGL nur Nullstellen mit negativen Realteilen, dann gehört diese Grundlösung g zu S' und man kann sie mit jeder kausalen Distribution f auf der rechten Seite der DGL falten. Gehört f überdies zu S' ; dann gilt auch der Faltungssatz der Fouriertransformation, d.h. es gilt

$$F(g * f) = F(g) * F(f).$$

Begründung: Die FT der kausalen Grundlösung gehört dann zum Raum der Multiplikatoren in S' .

Fazit: Die Faltung $g * f$ löst dann das inhomogene Problem $P(D)u=f$, $F(g)$ ist der Frequenzgang des Übertragungssystems, das durch die DGL beschrieben wird, und $F(g * f)$, das Spektrum der Lösung erhält man aus dem Spektrum der rechten Seite f einfach durch Multiplikation mit dem Frequenzgang.

Konkret könnten wir die DGL $u' + u = \sin(t)$ Heaviside(t) heranziehen.

Die DGL beschreibt ein Tiefpassfilter 1. Ordnung. Wir setzen einen energielosen Ruhezustand voraus und berechnen seine Reaktion auf das Aufschalten einer Sinusschwingung ab $t=0$.

Wir berechnen die kausale Grundlösung und deren Faltung mit der rechten Seite. Am Ergebnis sehen wir, dass die DGL stabil ist. Für $t \rightarrow \infty$ schwingt die Lösung wie $\cos(t) + \sin(t)$. Anschließend berechnen wir den Frequenzgang $F(g)$ und verifizieren wir damit den Faltungssatz.

```
In[89]:=
g[t_] = DSolve[ u'[t] + u[t] == DiracDelta[t], u[t], t][[1, 1, 2]] /.
C[1] -> 0 (* kausale Grundlösung g *)
```

```
Out[89]=  $e^{-t} \text{HeavisideTheta}[t]$ 
```

```
In[90]:= freggang[ω_] = FourierTransform[ g[t], t, ω, FourierParameters → {1, -1}]
(* der Frequenzgang *)
```

$$\text{Out[90]} = -\frac{i}{-i + \omega}$$


```
In[91]:= rs[t_] = Sin[t] HeavisideTheta[t]
(* die gewählte rechte Seite der DGL ab t=0 *)
loesunginhomogen[t_] = Convolve[g[s], rs[s], s, t]
(* die Lösung als Faltung von g und rs *)
```

```
Out[91]= HeavisideTheta[t] Sin[t]
```

$$\text{Out[92]} = -\frac{1}{2} \text{HeavisideTheta}[t] (\text{Cos}[t] - \text{Cosh}[t] - \text{Sin}[t] + \text{Sinh}[t])$$

Die Lösung ab $t=0$ ist also eine in der Amplitude gedämpfte und phasenverschobene Schwingung $\frac{1}{2}(\sin(t)-\cos(t))$ gleicher Frequenz wie bei der rechten Seite, überlagert vom schnell abklingenden Einschwingvorgang $\frac{1}{2}(\cosh(t) - \sinh(t)) = \frac{1}{2} e^{-t}$.

```
In[93]:= Plot[loesunginhomogen[t], {t, -2, 30},
PlotRange → {-1.5, 1.5}, PlotStyle → Directive[
Blue, Thickness[0.006]]]
```



Nun die Fouriertransformation der Lösung und anschließend das Produkt des Frequenzgangs mit der Fouriertransformierten der Inhomogenität. Wir erhalten beide Male das gleiche Ergebnis. Die Dirac-Impulse in der von *Mathematica* angegebenen Lösung entfallen insgesamt, wenn man mit dem Distributionenkalkül berücksichtigt, dass für C^∞ -Funktionen f gilt:

$$f(\omega) \cdot \delta(\omega - A) = f(A) \cdot \delta(\omega - A).$$

```
In[94]:= ftloesung[ω_] = Together [
FourierTransform[ loesunginhomogen[t], t, ω, FourierParameters → {1, -1}]]
```

$$\text{Out[94]} = -\frac{1}{(-i + \omega)(-1 + \omega^2)} \left(\frac{1}{4} + \frac{i}{4} \right) \left((-2 - 2i) + i\pi \text{DiracDelta}[-1 + \omega] - \pi\omega \text{DiracDelta}[-1 + \omega] - \right. \\ \left. i\pi\omega^2 \text{DiracDelta}[-1 + \omega] + \pi\omega^3 \text{DiracDelta}[-1 + \omega] + \pi \text{DiracDelta}[1 + \omega] + \right. \\ \left. i\pi\omega \text{DiracDelta}[1 + \omega] - \pi\omega^2 \text{DiracDelta}[1 + \omega] - i\pi\omega^3 \text{DiracDelta}[1 + \omega] \right)$$

```
In[95]:= fourrs[ω_] := FourierTransform[rs[t], t, ω, FourierParameters -> {1, -1}]
prodgf[ω_] = Together[freqgang[ω] fourrs[ω]]
```

```
Out[96]= (2 i + π DiracDelta[-1 + ω] - π ω^2 DiracDelta[-1 + ω] -
π DiracDelta[1 + ω] + π ω^2 DiracDelta[1 + ω]) / (2 (-1 + ω) (-i + ω) (1 + ω))
```

Wir testen auf Gleichheit :

```
In[97]:= test[ω_] = Simplify[ftloesung[ω] - prodgf[ω]]
```

```
Out[97]= 0
```

Faltungen und Fouriertransformation für Impulsfolgen

Ein großer Vorteil der Distributionentheorie ist, dass man mit ihr auch diskrete Systeme behandeln kann und dafür die mathematischen Hilfsmittel alle schon zur Verfügung hat. Eine diskrete Folge mit Folgengliedern c_k , $k \in \mathbb{Z}$, wird nun mit Dirac-Impulsen als Impulsfolge $\sum_{k=-\infty}^{+\infty} c_k \delta(t - ka)$ modelliert. Die Größe a ist dabei der als konstant betrachtete Zeitabstand, welcher der Abtastrate eines diskreten Systems entspricht. Alle benötigten Aussagen über das zugehörige diskrete System können dann mit den schon erlernten Methoden der Distributionentheorie hergeleitet werden.

Wie wir oben in **2.2.1 Beispiel 1** schon gesehen hatten, ist die Fouriertransformierte der Impulsfolge $\sum_{k=-\infty}^{+\infty} c_k \delta(t - ka)$ einfach eine verallgemeinerte Fourierreihe mit den Koeffizienten c_k :

$$F \left(\sum_{k=-\infty}^{+\infty} c_k \delta(t - ka) \right) = \sum_{k=-\infty}^{+\infty} c_k e^{-i k \omega a}.$$

In *Mathematica* können wir für die Faltung, sofern sie existiert, einfach den Befehl `DiscreteConvolve` für die Koeffizientenfolge verwenden, um eine Faltung zu berechnen.

Für Faltung zweier Impulsfolgen $\sum_{k=-\infty}^{+\infty} a_k \delta(t - ka) * \sum_{k=-\infty}^{+\infty} b_k \delta(t - ka)$ ergibt sich wieder - sofern die Faltung möglich ist - die Impulsfolge $\sum_{k=-\infty}^{+\infty} h_k \delta(t - ka)$ mit

$$h_k = \sum_{m=-\infty}^{+\infty} a_m b_{k-m}.$$

Die Fouriertransformierte der Faltung ist dann die verallgemeinerte Fourierreihe mit den Koeffizienten h_k .

Nachfolgend Bedingungen, unter denen die Faltung zweier Impulsfolgen T und G möglich ist und auch der Faltungssatz für die Fouriertransformation gilt:

- a) T hat schnell fallende Koeffizienten a_k , d.h. $|k|^m a_k \rightarrow 0$ für jedes $m \in \mathbb{N}$ und $|k| \rightarrow \infty$, und G hat polynomial beschränkte Koeffizienten b_k , d.h. $|b_k| \leq C |k|^m$ für alle $k \neq 0$ mit einer geeigneten Konstante C und geeignetem m.
- b) T hat absolut-summierbare Koeffizienten und G hat beschränkte Koeffizienten.

- c) T und G haben absolut-summierbare Koeffizienten.
 d) T und G haben quadratisch summierbare Koeffizienten.

Beweise dieser Aussagen findet man in [1], Kap. 9.

Ich gebe nur ein Beispiel mit *Mathematica*, interessierte Leser mögen selbst ein paar weitere ausprobieren. In einem nachfolgenden Notebook zu einige Grundlagen der diskreten Signalverarbeitung komme ich auf die hier dargelegten Sachverhalte zurück.


Beispiel 3. Wir hatten das Beispiel schon in meinem Notebook über Anwendungsbeispiele zur Distributionentheorie benutzt.

```
In[98]:= f1[n_] = 1/2^n UnitStep[n]
f2[n_] = 1/3^n UnitStep[n]
DiscreteConvolve[f1[n], f2[n], n, k, Assumptions -> k > 0]
DiscretePlot[3 x 2^-k - 2 x 3^-k, {k, 1, 20}, PlotRange -> All, PlotStyle -> Directive[
  Blue, Thickness[0.006]]]
```

Out[98]= $2^{-n} \text{UnitStep}[n]$

Out[99]= $3^{-n} \text{UnitStep}[n]$

Out[100]= $3 \times 2^{-k} - 2 \times 3^{-k}$



```
In[102]:= falt[n_] = DiscreteConvolve[f1[m], f2[m], m, n, Assumptions -> k > 0]
```

Out[102]=
$$\begin{cases} 3 \times 2^{-n} - 2 \times 3^{-n} & n \geq 0 \\ 0 & \text{True} \end{cases}$$


```
In[103]:= Sum[Piecewise[{{3 x 2^-n - 2 x 3^-n, n >= 0}}, 0], {n, 1, Infinity}]
```

Out[103]= 2

Die Koeffizienten der Faltung sind absolut-summierbar und die Fourierreihe mit diesen Koeffizienten als FT der Faltung ist eine klassische stetige $2\pi/a$ -periodische \mathbb{C} -wertige Funktion. Die Zeitkonstante a (in der Praxis der Zeitabstand zwischen zwei "Samples" in einem diskret arbeitenden System) können wir dabei frei wählen, z.B. $a=1$ [s]. Nachfolgend noch die Grafik einer Partialsumme der resultierenden Fourierreihe. **Solche Fourierreihen treten in**

den Anwendungen mit diskreten linearen Filtern als periodische Frequenzgänge der Filter auf.

```
In[104]:= a := 1
Plot[Abs[Sum[falt[n] Exp[-i n ω a], {n, 0, 20}]],
{ω, -4 π, 4 π}, PlotStyle → Directive[
Blue, Thickness[0.006]]]
```



3. Fouriertransformation für Funktionen und Distributionen bei mehrdimensionalem Parameterraum

Zum Abschluss sei gesagt, dass alles hier Gesagte natürlich auch auf den Fall mit einem mehrdimensionalen Parameterraum verallgemeinert werden kann. Gerade auch in der Theorie der linearen partiellen Differentialgleichungen mit konstanten Koeffizienten ist die Fouriertransformation ein mächtiges Werkzeug. Schließlich war die Lösung solcher Gleichungen für J. B. Fourier 1807 einmal der Anlass, diese Transformation in die Welt zu bringen. Ich werde auf dieses riesige Thema hier (vorläufig jedenfalls) aber nicht weiter eingehen, sondern dies mathematisch und im Umgang mit *Mathematica* erfahreneren Autoren auf diesem Gebiet überlassen.

Als Anregung seien daher nur noch zwei Beispiele von Fouriertransformationen für Funktionen mit 2 Variablen gegeben, die man der **Fourier-Optik** zuordnen kann. Mehr Details zu diesem Thema finden Sie bei Interesse in der Referenz [3] oder in spezieller Literatur zur Fourier-Optik.

Beispiel 1. (Die Rechteckblende) Wird kohärentes Licht mit der Amplitudenverteilung f an einer Blende der (x,y) -Ebene gebeugt, dann ist im Fall der Fraunhoferschen Beugung die Intensitätsverteilung des gebeugten Lichts auf einem Schirm proportional zu $|F(f)|^2$. Wir berechnen mit *Mathematica* und plotten $|F(f)|^2$ für die Rechteckfunktion $f = r[1,x] r[1,y]$. Im Plot kann man also "die Fouriertransformation sehen": Je höher der Wert im Bild desto heller erscheint der zugehörige Punkt im Beugungsbild. Das wohl aus dem Physikunterricht bekannte Beugungsmuster ist zu erkennen. Wir plotten $|F(f)|^2$ und den zugehörigen



DensityPlot.

```
In[106]:= f[x_, y_] = r[1, x] r[1, y]
fourxy[s_, t_] =
  Simplify[FourierTransform[f[x, y], {x, y}, {s, t}, FourierParameters -> {1, -1}]]
```

```
Out[106]= (-UnitStep[-1 + x] + UnitStep[1 + x]) (-UnitStep[-1 + y] + UnitStep[1 + y])
```

```
Out[107]=  $\frac{4 \sin[s] \sin[t]}{s t}$ 
```


```
In[108]:= Plot3D[Abs[fourxy[s, t]]^2, {s, -8, 8}, {t, -8, 8}, PlotRange -> All,
  RegionFunction -> Function[{x, y, z}, x^2 + y^2 != 0]] (*RegionFunction hier,
  um Division durch Null im Nullpunkt auszuschließen. Andernfalls
  meldet Mathematica mitunter einen Fehler und plottet nicht *)
DensityPlot[fourxy[s, t], {s, -8, 8}, {t, -8, 8}]
```



Beispiel 2. (Kreisblende) Hier betrachten wir die Fouriertransformation einer Kreisblende und setzen $f_2(x,y) = 1$ für $x^2 + y^2 \leq 1$.

```
In[110]:= f2[x_, y_] = Piecewise[{{1, x^2 + y^2 <= 1}, {0, True}}]
Plot3D[f2[x, y], {x, -2, 2}, {y, -2, 2}, PlotRange -> All, ViewPoint -> {3, 3, 1.5}]
```

```
Out[110]= { 1 x^2 + y^2 <= 1
           0 True
```



Leider erfolglos (bisher jedenfalls für mich) ist der Versuch einfach mit
`FourierTransform[f2[x,y],{x,y},{s,t}]`
 die Frage nach der FT an *Mathematica* zu richten.

Wenn man die Fouriertransformierte selbst berechnet (als Hinweis dazu siehe ganz unten die letzten Zeilen), ergibt sich wegen der Rotationssymmetrie von f_2 auch eine rotationssymmetrische Fouriertransformierte. In Polarkoordinaten ρ und ϕ ist diese also unabhängig von ϕ und lautet hier mit $R=1$ und der Funktion f_2 wie oben

$$F(f_2)(\rho \cos(\phi), \rho \sin(\phi)) = F(f_2)(\rho) = 2\pi \int_0^1 J_0(r\rho) r dr.$$

Dabei ist J_0 die **Besselfunktion der Ordnung Null**. Speziell ist $F(f_2)(0)$ der Flächeninhalt der verwendeten Kreisscheibe, hier also gleich π .


```
In[112]:= fourf2[ρ_] = 2 π Integrate[BesselJ[0, r ρ] r, {r, 0, 1}]
(* FT von f2 in Polarkoordinaten *)
Limit[fourf2[ρ], ρ -> 0] (* der Grenzwert für ρ->0 ist die Kreisflächeninhalt,
hier des Einheitskreises *)
```

```
Out[112]=  $\frac{2 \pi \text{BesselJ}[1, \rho]}{\rho}$ 
```

```
Out[113]=  $\pi$ 
```


```
In[114]:= Plot[Abs[fourf2[ρ]]2, {ρ, -10, 10}, PlotRange → All]
(* die FT als Funktion nur von ρ *)
```

Out[114]=




```
In[115]:= ParametricPlot3D[{ρ Cos[φ], ρ Sin[φ], Abs[fourf2[ρ]]2}, {ρ, 0, 10},
{φ, 0, 2 π}, PlotRange → All] (* hier die Rotation der Kurve *)
```

Out[115]=



In[116]:= `DensityPlot[Abs[fourf2[r]]^2, {r, 0, 4}, {φ, 0, 2π}]` (* `DensityPlot`,
 der Einheitskreis als (r, ϕ) -Rechteck dargestellt, auch hier ist die
 Unabhängigkeit von ϕ an der "Streifenstruktur" gut zu erkennen*)



Hier noch kurz, wie man mit *Mathematica* die gesuchte FT findet:

Man verwende Polarkoordinaten statt kartesischer im Ausgangs- und im Ziel-Parameterraum, also $x=r\cos(\phi)$, $y=r\sin(\phi)$ als Ausgangsparameter und $x_1 = \rho \cos(\psi)$, $x_2 = \rho \sin(\psi)$ als Parameter der Fouriertransformierten. Wenn Sie das hinschreiben und die Substitution $\theta = \phi - \psi - \pi/2$ im Integral verwenden, erhalten Sie das Ergebnis. Es ist also von Vorteil, wenn man trotz aller Power, die *Mathematica* bietet, bisweilen auch noch selbst mathematisch eingreifen kann.

In[117]:= `f0[r_] = r[1, r]`
`fourf2polar[x1, x2] =`
`Assuming[ρ > 0, Integrate[Integrate[f0[r] ei r ρ sin[θ] r, {θ, 0, 2π}], {r, 0, ∞}]]`

Out[117]= `-UnitStep[-1 + r] + UnitStep[1 + r]`

Out[118]=
$$\frac{2\pi \text{BesselJ}[1, \rho]}{\rho}$$

Literaturhinweise : Bei mehr Interesse an Distributionen empfehle ich neben [1]

[2] H. Triebel

Höhere Analysis

[3] J.S. Walker

Fourier Analysis, Oxford Univ. Press 1988

[4] A.H. Zemanian

Distribution Theory And Transform Analysis,
 Dover 1965

[5] V.S. Vladimirov

Methods of the Theory of Generalized Functions
 Taylor and Francis, 2002